

# Efficient Mendler-Style Lambda-Encodings in Cedille

Denis Firsov, Richard Blair, and Aaron Stump

Department of Computer Science  
The University of Iowa  
Iowa City, IA, USA  
{name-lastname}@uiowa.edu

**Abstract.** It is common to model inductive datatypes as least fixed points of functors. We show that within the Cedille type theory we can relax functoriality constraints and generically derive an induction principle for Mendler-style lambda-encoded inductive datatypes, which arise as least fixed points of covariant schemes where the morphism lifting is defined only on identities. Additionally, we implement a destructor for these lambda-encodings that runs in constant-time. As a result, we can define lambda-encoded natural numbers with an induction principle and a constant-time predecessor function so that the normal form of a numeral requires only linear space. The paper also includes several more advanced examples.

**Keywords:** type theory, lambda-encodings, Cedille, induction principle, predecessor function, inductive datatypes

## 1 Introduction

It is widely known that inductive datatypes may be defined in pure impredicative type theory. For example, Church encodings identify each natural number  $n$  with its iterator  $\lambda s. \lambda z. s^n z$ . The Church natural numbers can be typed in System F by means of impredicative polymorphism:

```
cNat ◀ * = ∀ X : *. (X → X) → X → X.
```

The first objection to lambda-encodings is that it is provably impossible to derive an induction principle in second-order dependent type theory [1]. As a consequence, most languages come with a built-in infrastructure for defining inductive datatypes and their induction principles. Here are the definitions of natural numbers in Agda and Coq:

```
data Nat : Set                                Inductive nat : Type :=
  zero : Nat                                   | 0 : nat
  suc  : Nat → Nat                             | S : nat → nat.
```

Coq will automatically generate the induction principle for `nat`, and in Agda it can be derived by pattern matching and explicit structural recursion.

Therefore, we can ask if it is possible to extend the Calculus of Constructions with **typing constructs** that make induction derivable for lambda-encoded datatypes. Stump gave a positive answer to this question by introducing the Calculus of Dependent Lambda Eliminations (CDLE) [2]. CDLE is a Curry-style Calculus of Constructions extended with implicit products, intersection types, and primitive heterogeneous equality. Stump proved that natural number induction is derivable in this system for lambda-encoded natural numbers. Later, we generalized this work by deriving induction for lambda-encodings of inductive datatypes which arise as least fixed points of functors [3]. Moreover, we observed that the proof of induction for Mendler-style lambda-encoding relied only on the identity law of functors. In this paper, we exploit this observation to define a new class of covariant schemes, which includes functors, and induces a larger class of inductive datatypes supporting derivable induction.

Another objection to lambda-encodings is their computational inefficiency. For example, computing the predecessor of a Church encoded Peano natural provably requires linear time [4]. The situation was improved by Parigot who proposed a new lambda-encoding of numerals with a constant-time predecessor, but the size of the number  $n$  is exponential  $O(2^n)$  [5]. Later, the situation was improved further by the Stump-Fu encoding, which supports a constant-time predecessor and reduces the size of the natural number  $n$  to  $O(n^2)$  [6]. In this paper, we show how to develop a constant-time predecessor within CDLE for a Mendler-style lambda-encoded naturals that are linear in space.

This paper makes the following technical contributions:

1. We introduce a new kind of parameterized scheme using **identity mappings** (function lifting defined only on identities). We show that every functor has an associated identity mapping, but not vice versa.
2. We use a Mendler-style lambda-encoding to prove that every scheme with an identity mapping induces an inductive datatype. Additionally, we generically derive an induction principle for these datatypes.
3. We implement a generic constant-time destructor of Mendler-style lambda-encoded inductive datatypes. To the best of our knowledge, we offer a first example of typed lambda-encoding of inductive datatypes with derivable induction and a constant-time destructor where normal forms of data require linear space.
4. We give several examples of concrete datatypes defined using our development. We start by giving a detailed description of lambda-encoded naturals with an induction principle and a constant-time predecessor function that only requires linear space to encode a numeral. We also give examples of infinitary datatypes. Finally, we present an inductive datatype that arises as a least fixed point of a scheme that is not a functor, but has an identity mapping.

## 2 Background

In this section, we briefly summarize the main features of Cedille’s type theory. For full details on CDLE, including semantics and soundness results, please see the previous papers [2, 7]. The main metatheoretic property proved in the previous work is logical consistency: there are types which are not inhabited. CDLE is an extrinsic (i.e., Curry-style) type theory, whose terms are exactly those of the pure untyped lambda calculus (with no additional constants or constructs). The type-assignment system for CDLE is not subject-directed, and thus cannot be used directly as a typing algorithm. Indeed, since CDLE includes Curry-style System F as a subsystem, type assignment is undecidable [8]. To obtain a usable type theory, Cedille thus has a system of annotations for terms, where the annotations contain sufficient information to type terms algorithmically. But true to the extrinsic nature of the theory, these annotations play no computational role. Indeed, they are erased both during compilation and before formal reasoning about terms within the type theory, in particular by definitional equality (see Figure 1).

$$\begin{array}{c}
 \frac{\Gamma, x : T' \vdash t : T \quad x \notin FV(|t|)}{\Gamma \vdash \Lambda x : T'. t : \forall x : T'. T} \quad \frac{\Gamma \vdash t : \forall x : T'. T \quad \Gamma \vdash t' : T'}{\Gamma \vdash t - t' : [t'/x]T} \\
 \\
 \frac{\Gamma \vdash t : T}{\Gamma \vdash \beta : t \simeq t} \quad \frac{\Gamma \vdash t' : t_1 \simeq t_2 \quad \Gamma \vdash t : [t_1/x]T}{\Gamma \vdash \rho t' - t : [t_2/x]T} \quad \begin{array}{l} |\Lambda x : T. t| = |t| \\ |t - t'| = |t| \\ |\beta| = \lambda x. x \\ |\rho t' - t| = |t| \\ |[t_1, t_2\{p\}]| = |t_1| \\ |t.1| = |t| \\ |t.2| = |t| \end{array} \\
 \\
 \frac{\Gamma \vdash t_1 : T \quad \Gamma \vdash t_2 : [t_1/x]T' \quad \Gamma \vdash p : |t_1| \simeq |t_2|}{\Gamma \vdash [t_1, t_2\{p\}] : \iota x : T. T'} \\
 \\
 \frac{\Gamma \vdash t : \iota x : T. T'}{\Gamma \vdash t.1 : T} \quad \frac{\Gamma \vdash t : \iota x : T. T'}{\Gamma \vdash t.2 : [t.1/x]T'}
 \end{array}$$

**Fig. 1.** Introduction, elimination, and erasure rules for additional type constructs

CDLE extends the (Curry-style) Calculus of Constructions (CC) with primitive heterogeneous equality, intersection types, and implicit products:

- $\mathfrak{t}_1 \simeq \mathfrak{t}_2$ , a heterogeneous equality type. The terms  $\mathfrak{t}_1$  and  $\mathfrak{t}_2$  are required to be typed, but need not have the same type. We introduce this with a constant  $\beta$  which erases to  $\lambda x. x$  (so our type-assignment system has no additional constants, as promised);  $\beta$  proves  $\mathfrak{t} \simeq \mathfrak{t}$  for any typeable term  $\mathfrak{t}$ . Combined with definitional equality,  $\beta$  proves  $\mathfrak{t}_1 \simeq \mathfrak{t}_2$  for any  $\beta\eta$ -equal  $\mathfrak{t}_1$  and  $\mathfrak{t}_2$  whose free variables are all declared in the typing context. We eliminate the equality type by rewriting, with a construct  $\rho \mathfrak{t}' - \mathfrak{t}$ . Suppose  $\mathfrak{t}'$  proves  $\mathfrak{t}_1 \simeq \mathfrak{t}_2$  and we synthesize a type  $T$  for  $\mathfrak{t}$ , where  $T$  has several occurrences of terms definitionally equal to  $\mathfrak{t}_1$ . Then the type synthesized for  $\rho \mathfrak{t}' - \mathfrak{t}$

is  $T$  except with those occurrences replaced by  $t_2$ . Note that the types of the terms are not part of the equality type itself, nor does the elimination rule require that the types of the left-hand and right-hand sides are the same to do an elimination.

- $\iota x : T. T'$ , the dependent intersection type of Kopylov [9]. This is the type for terms  $t$  which can be assigned both the type  $T$  and the type  $[t/x]T'$ , the substitution instance of  $T'$  by  $t$ . There are constructs  $t.1$  and  $t.2$  to select either the  $T$  or  $[t.1/x]T'$  view of a term  $t$  of type  $\iota x : T. T'$ . We introduce a value of  $\iota x : T. T'$  by construct  $[t_1, t_2 \{p\}]$ , where  $t_1$  has type  $T$  (algorithmically),  $t_2$  has type  $[t_1/x]T'$ , and  $p$  proves  $t_1 \simeq t_2$ .
- $\forall x : T. T'$ , the implicit product type of Miquel [10]. This can be thought of as the type for functions which accept an erased input of type  $x : T$ , and produce a result of type  $T'$ . There are term constructs  $\Lambda x. t$  for introducing an implicit input  $x$ , and  $t \text{-} t'$  for instantiating such an input with  $t'$ . The implicit arguments exist just for purposes of typing so that they play no computational role and equational reasoning happens on terms from which the implicit arguments have been erased.

It is important to understand that the described constructs are erased before the formal reasoning, according to the erasure rules in Figure 1.

We have implemented CDLE in a tool called Cedille, which we have used to typecheck the developments of this paper. The pre-release version of Cedille and the Cedille code accompanying this paper could be found here:

<http://firsov.ee/efficient-lambda/>

### 3 Preliminaries

We skip the details of the lambda-encoded implementation of basic datatypes like `Unit`, `Empty`, sums  $(X + Y)$ , and dependent sums  $(\Sigma x : X. Y x)$ , for which the usual introduction and elimination rules are derivable in Cedille.

In this paper, we use syntactical simplifications to improve readability. In particular, we hide the type arguments in the cases when they are unambiguous. For example, if  $x : X$  and  $y : Y$  then we write `pair x y` instead of fully type-annotated `pair X Y x y`. The current version of Cedille requires fully annotated terms.

#### 3.1 Multiple Types of Terms

CDLE's dependent intersection types allow propositionally equal values to be intersected. Given  $x : X$ ,  $y : Y$ , and a proof  $p$  of  $x \simeq y$ , we can introduce an intersection value  $v := [x, y \{p\}]$  of type  $\iota x : X. Y x$ . Every intersection has two "views": the first view  $v.1$  has type  $X$  and the second view  $v.2$  has type  $Y v.1$ . The term  $[x, y \{p\}]$  erases to  $x$  according to the erasure rules in Figure 1. This allows us to see  $x$  as having two distinct types, namely  $X$  and  $Y x$ :

$$\begin{aligned} \text{phi} &\triangleleft \forall X: \star. \forall Y: X \rightarrow \star. \Pi x: X. \forall y: Y\ x. \forall p: x \simeq y. Y\ x \\ &= \Lambda X. \Lambda Y. \lambda x. \Lambda y. \Lambda p. [x, y\ \{p\}].2. \end{aligned}$$

Indeed, the definition of `phi` erases to the term  $\lambda x. x$ . Hence, `phi`  $x$   $-y$   $-p$  beta-reduces to  $x$  and has type  $Y\ x$  (dash denotes the application of implicitly quantified arguments). Note that  $\Pi x : X. T$  is the usual “explicit” dependent function space.

In our setting, we implemented the `phi` combinator by utilizing propositional intersection types that allow us to intersect provably equal terms. The converse is also possible: `phi` can be taken as a language primitive and then used to derive propositional intersection types from definitional intersection types, where only definitionally equal values are allowed to be intersected (this variation of CDLE is described and used in [11]).

### 3.2 Identity Functions

In our setting, it is possible to implement a function of type  $X \rightarrow Y$  so that it erases to term  $\lambda x. x$  where  $X$  is different from  $Y$ . The simplest example is the first (or second) “view” from an intersection value:

$$\begin{aligned} \text{view1} &\triangleleft \forall X : \star. \forall Y : X \rightarrow \star. (\iota\ x : X. Y\ x) \rightarrow X \\ &= \Lambda X. \Lambda Y. \lambda x. x.1. \end{aligned}$$

Indeed, according to the erasure rules `view1` erases to the term  $\lambda x. x$ . We introduce a type `Id`  $X$   $Y$ , which is the set of all functions from  $X$  to  $Y$  that erase to the identity function  $(\lambda x. x)$ :

$$\text{id} \triangleleft \forall X : \star. X \rightarrow X = \Lambda X. \lambda x. x.$$

$$\text{Id} \triangleleft \star \rightarrow \star \rightarrow \star = \lambda X. \lambda Y. \Sigma f : X \rightarrow Y. f \simeq \text{id}.$$

*Introduction* The importance of the previously implemented combinator `phi` is that it allows to introduce an identity function `Id`  $X$   $Y$  from any extensional identity  $f : X \rightarrow Y$  (i.e.,  $x \simeq f\ x$  for any  $x$ ):

$$\begin{aligned} \text{intrId} &\triangleleft \forall X\ Y : \star. \Pi f : X \rightarrow Y. (\Pi x : X. x \simeq f\ x) \rightarrow \text{Id}\ X\ Y \\ &= \Lambda X. \Lambda Y. \lambda f. \lambda \text{prf}. \text{pair}\ (\lambda x. \text{phi}\ x\ -(f\ x)\ -(\text{prf}\ x))\ \beta. \end{aligned}$$

*Elimination* The identity function `Id`  $X$   $Y$  allows us to see values of type  $X$  as also having type  $Y$ . The implementation is in terms of `phi` combinator:

$$\begin{aligned} \text{elimId} &\triangleleft \forall X\ Y : \star. \forall c : \text{Id}\ X\ Y. X \rightarrow Y = \\ &= \Lambda X. \Lambda Y. \Lambda c. \lambda x. \text{phi}\ x\ -(\pi_1\ c\ x)\ -(\rho\ (\pi_2\ c)\ -\beta). \end{aligned}$$

The subterm  $\rho\ (\pi_2\ c)\ -\beta$  proves  $x \simeq \pi_1\ c\ x$ , where  $\pi_i$  is the  $i$ -th projections from a dependent sum. Observe that `elimId` itself erases to  $\lambda x. x$ , hence `elimId`  $-c\ x \simeq x$  by beta-reduction. In other words, an identity function `Id`  $X$   $Y$  allows  $x : X$  to be seen as having types  $X$  and  $Y$  at the same time.

### 3.3 Identity Mapping

A scheme  $F : \star \rightarrow \star$  is a functor if it comes equipped with a function `fmap` that satisfies the identity and composition laws:

```

Functor ◀ ( $\star \rightarrow \star$ )  $\rightarrow \star = \lambda F$ .
   $\Sigma \text{fmap} : \forall X Y : \star. (X \rightarrow Y) \rightarrow F X \rightarrow F Y$ .
  IdentityLaw fmap  $\times$  CompositionLaw fmap.

```

However, it is simple to define a covariant scheme for which the function `fmap` cannot be implemented (below,  $x_1 \neq x_2$  is shorthand for  $x_1 \simeq x_2 \rightarrow \text{Empty}$ ):

```

UneqPair ◀  $\star \rightarrow \star = \lambda X. \Sigma x_1 : X. \Sigma x_2 : X. x_1 \neq x_2$ .

```

We introduce schemes with *identity mappings* as a new class of parameterized covariant schemes. An identity mapping is a lifting of identity functions:

```

IdMapping ◀ ( $\star \rightarrow \star$ )  $\rightarrow \star = \lambda F$ .
   $\forall X Y : \star. \text{Id } X Y \rightarrow \text{Id } (F X) (F Y)$ .

```

Intuitively, `IdMapping F` is similar to `fmap` of functors, but it needs to be defined only on identity functions. The identity law is expressed as a requirement that an identity function of type `Id X Y` is mapped to an identity function of type `Id (F X) (F Y)`.

Clearly, every functor induces an identity mapping (by the application of `intrId` to `fmap` and its identity law):

```

fm2im ◀  $\forall F : \star \rightarrow \star. \text{Functor } F \rightarrow \text{IdMapping } F = \langle \dots \rangle$ 

```

However, `UneqPair` is an example of scheme which is not a functor, but has an identity mapping (see example in Section 6.3).

In the rest of the paper we show that every identity mapping `IdMapping F` induces an inductive datatype which is a least fixed point of `F`. Additionally, we generically derive an induction principle and implement a constant-time destructor for these datatypes.

## 4 Inductive Datatypes from Identity Mappings

In our previous paper, we used Cedille to show how to generically derive an induction principle for Mendler-style lambda-encoded datatypes that arise as least fixed points of functors [3]. In this section, we revisit this derivation to show that it is possible to relax functoriality constraints and only assume that the underlying signature scheme is accompanied by an identity mapping.

### 4.1 Basics of Mendler-Style Encoding

In this section, we investigate the standard definitions of Mendler-style `F`-algebras that are well-defined for any unrestricted scheme  $F : \star \rightarrow \star$ . To reduce the notational clutter, we assume that  $F : \star \rightarrow \star$  is a global (module) parameter:

```
module _ (F : * → *).
```

In the abstract setting of category theory, a Mendler-style F-algebra is a pair  $(X, \Phi)$  where  $X$  is an object (i.e., the *carrier*) in  $\mathcal{C}$  and  $\Phi : \mathcal{C}(-, X) \rightarrow \mathcal{C}(F -, X)$  is a natural transformation. In the concrete setting of Cedille, objects are types, arrows are functions, and natural transformations are polymorphic functions. Therefore, Mendler-style F-algebras are defined as follows:

```
AlgM ◀ * → * = λ X. ∀ R : *. (R → X) → F R → X.
```

Uustalu and Vene showed that initial Mendler-style F-algebras offer an alternative categorical model of inductive datatypes [12]. The carrier of an initial F-algebra is an inductive datatype that is a least fixed point of F. It is known that if F is a positive scheme then the least fixed point of it may be implemented in terms of universal quantification [13]:

```
FixM ◀ * = ∀ X : *. AlgM X → X.
```

```
foldM ◀ ∀ X : *. AlgM X → FixM → X = Λ X. λ alg. λ x. x alg.
```

In essence, this definition identifies inductive datatypes with iterators and every function on `FixM` is to be computed by iteration.

The natural transformation of the initial Mendler-style F-algebra denotes the collection of constructors of its carrier [12]. In our setting, the initial Mendler-style F-algebra `AlgM FixM` is not definable because F is not a functor [3]. Instead, we express the collection of constructors of datatype `FixM` as a conventional F-algebra  $F \text{ FixM} \rightarrow \text{FixM}$ :

```
inFixM ◀ F FixM → FixM = λ x. Λ X. λ alg. alg (foldM alg) x.
```

The function `inFixM` is of crucial importance because it expresses constructors of `FixM` without requirements of functoriality on  $F : * \rightarrow *$ .

It is provably impossible to define the mutual inverse of `inFixM` (destructor of `FixM`) without introducing additional constraints on F. Assume the existence of function `outFixM` (which need not be an inverse of `inFixM`), typed as follows:

```
outFixM ◀ ∀ F : * → *. FixM F → F (FixM F) = <..>
```

Next, recall that in the impredicative setting the empty type is encoded as  $\forall X : *. X$  (its inhabitant implies any equation). Then, we instantiate F with the negative polymorphic scheme  $\text{NegF } X := \forall Y : *. X \rightarrow Y$ , and exploit the function `outFixM` to construct a witness of the empty type:

```
T ◀ * = FixM NegF.
```

```
ty ◀ ∀ Y : *. T → Y = Λ Y. λ t. outFixM -NegF t -Y t.
```

```
t ◀ T = ty -T (inFixM -NegF ty).
```

```
unsound ◀ ∀ X : *. X = Λ X. ty -X t.
```

Hence, the existence of function `outFixM` contradicts the consistency of Cedille. Hence, the inverse of `inFixM` can exist only for some restricted class of schemes  $F : \star \rightarrow \star$ .

## 4.2 Inductive Subset

From this point forward we assume that the scheme  $F$  is also accompanied by an identity mapping `imap`:

```
module _ (F :  $\star \rightarrow \star$ ) (imap : IdMapping F).
```

In our previous work we assumed that  $F$  is a functor and showed how to specify the “inductive” subset of the type `FixM F`. Then, we generically derived induction for this subset. In this section, we update the steps of our previous work to account for  $F : \star \rightarrow \star$  not being a functor.

The dependent intersection type  $\iota x : X. Y x$  can be understood as a subset of  $X$  defined by a predicate  $Y$ . However, to construct the value of this type we must provide  $x : X$  and a proof  $p : Y x$  so that  $x$  and  $p$  are provably equal ( $x \simeq p$ ). Hence, to align with this constraint we use implicit products to express inductivity of `FixM` as its “dependently-typed” version. Recall that `FixM` is defined in terms of Mender-style  $F$ -algebras:

```
AlgM  $\triangleleft \star \rightarrow \star = \lambda X. \forall R : \star. (R \rightarrow X) \rightarrow F R \rightarrow X$ .
```

In our previous work, we introduced the *Q-proof F-algebras* as a “dependently-typed” counterpart of `AlgM`. The value of type `PrfAlgM X Q alg` should be understood as an inductive proof that predicate  $Q$  holds for every  $X$  where  $X$  is a least fixed point of  $F$  and `alg` :  $F X \rightarrow X$  is a collection of constructors of  $X$ .

```
PrfAlgM  $\triangleleft \Pi X : \star. (X \rightarrow \star) \rightarrow (F X \rightarrow X) \rightarrow \star$   
=  $\lambda X. \lambda Q. \lambda \text{alg}.$   
 $\forall R : \star. \forall c : \text{Id } R X. (\Pi r : R. Q (\text{elimId } -c r)) \rightarrow$   
 $\Pi fr : F R. Q (\text{alg } (\text{elimId } -(\text{imap } c) fr)).$ 
```

Mender-style  $F$ -algebras (`AlgM`) allow recursive calls to be explicitly stated by providing arguments  $R \rightarrow X$  and  $F R$ , where the polymorphically quantified type  $R$  ensures termination. Similarly,  $Q$ -proof  $F$ -algebras allow the inductive hypotheses to be explicitly stated for every  $R$  by providing an implicit identity function  $c : \text{Id } R X$ , and a dependent function of type  $\Pi r : R. Q (\text{elimId } -c r)$  (recall that `elimId -c r` reduces to  $r$  and has type  $X$ ). Given the inductive hypothesis for every  $R$ , the proof algebra must conclude that the predicate  $Q$  holds for every  $X$ , which is produced by constructors `alg` from any given  $F R$  that has been “casted” to  $F X$ .

Next, recall that `FixM` is defined as a function from `AlgM X` to  $X$  for every  $X$ .

```
FixM  $\triangleleft \star = \forall X : \star. \text{AlgM } X \rightarrow X$ .
```

To retain the analogy of definitions, we express the inductivity of value  $x : \text{FixM}$  as a dependent function from a  $Q$ -proof  $F$ -algebra to  $Q x$ .



```
IsIndFixM ◀ FixM → * = λ x : FixM.
  ∀ Q : FixM → *. PrfAlgm FixM Q inFixM → Q x.
```

Now, we employ intersection types to define a type `FixIndM` as a subset of `FixM` carved out by the “inductivity” predicate `IsIndFixM`:

```
FixIndM ◀ * = ι x : FixM. IsIndFixM x.
```

Finally, we must explain how to construct the values of this type. As in the case of `FixM`, the set of constructors of `FixIndM` is expressed by a conventional F-algebra `F FixIndM → FixIndM`. The implementation is divided into three steps:

First, we define a function from `F FixIndM` to `FixM`:

```
tc1 ◀ F FixIndM → FixM = λ x.
  let c ◀ Id (F FixIndM) (F FixM)
    = imap (intrId (λ x. x.1) (λ x. β)) in
  inFixM (elimId -c x).
```

The implementation simply “casts” its argument to `F FixM` and then applies the previously implemented constructor of `FixM` (`inFixM`). Because `elimId -c x` reduces to `x`, the erasure of `tc1` is the same as the erasure of `inFixM` which is a term `λ x. λ q. q (λ r. r q) x`.

Second, we show that the same lambda term could also be typed as a proof that every `tc1 x` is inductive:

```
tc2 ◀ Π x : F FixIndM. IsIndFixM (tc1 x) = λ x.
  (Λ Q. λ q. (q -(intrId (λ x. x.1) (λ x. β)) (λ r. r.2 q) x)).
```

Indeed, functions `tc1` and `tc2` are represented by the same pure lambda term.

Finally, given any value `x : F FixInd` we can intersect `tc1 x` and the proof of its inductivity `tc2 x` to construct an element of an inductive subset `FixIndM`:

```
inFixIndM ◀ F FixIndM → FixIndM = λ x. [ tc1 x, tc2 x { β } ].
```

Recall that erasure of intersection `[ x, y {p} ]` equals the erasure of `x`. Therefore, functions `inFixM`, `tc1`, `tc2`, and `inFixIndM` all erase to the same pure lambda term. In other words, in Cedille the term `λ x. λ q. q (λ r. r q) x` can be extrinsically typed as any of these functions.

### 4.3 Induction Principle

We start by explaining why we need to derive induction for `FixIndM`, even though it is definitionally an inductive subset of `FixM`. Indeed, every value `x : FixIndM` can be “viewed” as a proof of its own inductivity. More precisely, the term `x.2` is a proof of the inductivity of `x.1`. Moreover, the equational theory of CDLE gives us the following equalities `x.1 ≃ x ≃ x.2` (due to the rules of erasure). But recall that the inductivity proof provided by the second view `x.2` is typed as follows:

```
∀ Q : Fix → *. PrfAlgm FixM Q inFixM → Q x.1
```

Note that  $Q$  is a predicate on  $\text{FixM}$  and not  $\text{FixIndM}$ ! This form of inductivity does not allow properties of  $\text{FixIndM}$  to be proven.

Therefore, our goal is to prove that every  $x : \text{FixIndM}$  is inductive in its own right. We phrase this in terms of proof-algebras parameterized by  $\text{FixIndM}$ , a predicate on  $\text{FixIndM}$ , and its constructors ( $\text{inFixIndM}$ ):

$$\forall Q : \text{FixIndM} \rightarrow \star. \text{PrfAlgM } \text{FixIndM } Q \text{ inFixIndM} \rightarrow Q \ x.$$

In our previous work, we already made an observation that the derivation of induction for Mendler-style encodings relies only on the identity law of functors [3]. Therefore, the current setting only requires minor adjustments of our previous proof. For the sake of completeness, we present a main idea of this derivation.

The key insight is that we can convert predicates on  $\text{FixIndM}$  to logically equivalent predicates on  $\text{FixM}$  by using heterogeneous equality:

$$\begin{aligned} \text{Lift} &\triangleleft (\text{FixInd} \rightarrow \star) \rightarrow \text{Fix} \rightarrow \star \\ &= \lambda Q. \lambda y. \Sigma \ x : \text{FixInd}. \ x \simeq y \times Q \ x. \end{aligned}$$

$$\text{eqv1} \triangleleft \Pi \ x : \text{FixIndM}. \ \forall Q : \text{FixIndM} \rightarrow \star. \ Q \ x \rightarrow \text{Lift } Q \ x.1 = \langle \dots \rangle$$

$$\text{eqv2} \triangleleft \Pi \ x : \text{FixIndM}. \ \forall Q : \text{FixIndM} \rightarrow \star. \ \text{Lift } Q \ x.1 \rightarrow Q \ x = \langle \dots \rangle$$

These properties allow us to convert a  $Q$ -proof algebra to a proof algebra for a lifted predicate  $\text{Lift } Q$ , and then derive the generic induction principle:

$$\begin{aligned} \text{convIH} &\triangleleft \forall Q : \text{FixIndM} \rightarrow \star. \text{PrfAlgM } \text{FixIndM } Q \text{ inFixIndM} \\ &\rightarrow \text{PrfAlgM } \text{FixM } (\text{Lift } Q) \text{ inFixM} = \langle \dots \rangle \end{aligned}$$

$$\begin{aligned} \text{induction} &\triangleleft \forall Q : \text{FixIndM} \rightarrow \star. \text{PrfAlgM } \text{FixIndM } Q \text{ inFixIndM} \rightarrow \\ &\Pi \ e : \text{FixIndM}. \ Q \ e = \Lambda Q. \lambda p. \lambda e. \text{eqv2 } e \ (e.2 \ (\text{convIH } p)). \end{aligned}$$

Let  $Q$  be a predicate on  $\text{FixIndM}$  and  $p$  be a  $Q$ -proof algebra: we show that  $Q$  holds for any  $e : \text{FixIndM}$ . Recall that every  $e : \text{FixIndM}$  can be viewed as a proof of inductivity of  $e.1$  via  $e.2 : \text{IsIndFixM } e.1$ . We use this to get a proof of the lifted predicate  $\text{Lift } Q \ e.1$  from the proof algebra delivered by  $\text{convIH } p$ . Finally, we get  $Q \ e$  by using  $\text{eqv2}$ .

## 5 Constant-Time Destructors

An induction principle is needed to prove properties about programs, but practical functional programming also requires constant-time destructors (also called accessors) of inductive datatypes. Let us illustrate the problem using the datatype of natural numbers. In Agda it is easy to implement the predecessor function by pattern matching:

```
pred : Nat → Nat
pred zero = zero
pred (suc n) = n
```

The correctness of `pred` trivially follows by beta-reduction:

```
predProp : (n : Nat) → pred (suc n) ≡ n
predProp n = refl
```

Let us switch to Cedille and observe that it is much less trivial to implement the predecessor for the impredicative encoding of Peano numerals. Here is the definition of Church encoded Peano naturals and their constructors:

```
cNat ◀ * = ∀ X : *. (X → X) → X → X.

zero ◀ cNat = λ X. λ s. λ z. z.
suc ◀ cNat → cNat = λ n. λ X. λ s. λ z. s (n s z).
```

Next, we implement the predecessor for `cNat` which is due to Kleene:

```
zCase ◀ cNat × cNat = pair zero zero
sCase ◀ cNat × cNat → cNat × cNat
      = λ n. pair (π2 n) (suc (π2 n)).

predK ◀ Nat → Nat = λ n. π1 (n sCase zCase).
```

The key to the Kleene predecessor is the function `sCase`, which ignores the first item of the input pair, moves the second natural to the first position, and then applies the successor of the second element within the second position. Hence, folding a natural number `n` with `sCase` and `zCase` produces a pair `(n-1, n)`. In the end, `predK n` projects the first element of a pair.

Kleene predecessor runs in linear time. Also, `predK (suc n)` gets stuck after reducing to `π1 (pair (π2 (n sCase zCase)) (suc (π2 (n sCase zCase))))`. Hence, we must use induction to prove that `predK (suc n)` computes to `n`.

Furthermore, Parigot proved that any definition of predecessor for the Church-style lambda-encoded numerals requires linear time [4].

### 5.1 Constant-Time Destructor for Mendler-Style Encoding

In previous sections we defined a datatype `FixIndM` for every scheme `F` that has an identity mapping. Then, we implemented the constructors of the datatype as the function `inFixIndM`, and defined an induction principle phrased in terms of this function. In this section, we develop a mutual inverse of `inFixIndM` that runs in constant time. As a simple consequence, we prove that `FixIndM` is a least fixed point of `F`.

Let us start by exploring the computational behaviour of the function `foldM`. The following property is a variation of the *cancellation law* for Mendler-style encoded data [12], and its proof is simply by beta-reduction.

```
foldHom ◀ ∀ X : *. Π x : F FixM. Π alg : AlgM X.
      foldM alg (inFixM x) ≃ alg (foldM alg) x = λ X. λ x. λ a. β.
```

In other words, folding the inductive value `inFixM x` replaces its outermost “constructor” `inFixM` with the partially applied F-algebra `alg (foldM alg)`.

It is well-known that (computationally) induction can be reduced to iteration (folding). Therefore, we can state the cancellation law for the induction rule in terms of proof algebras.

```
indHom ◀ ∀ Q : FixIndM → *. Π alg : PrfAlgM FixIndM Q inFixIndM.
  Π x : F FixIndM. ∀ c : Id FixIndM FixIndM.
    induction alg (inFixInd x) ≃ alg -c (induction alg) x
  = Λ Q. λ p. λ x. β.
```

Crucially, the proof of `indHom` is by reflexivity ( $\beta$ ), which ensures that the left-hand side of equality beta-reduces to the right-hand side in a constant number of beta-reductions.

Next, we implement a proof algebra for the predicate  $\lambda \_ . F \text{ FixIndM}$ .

```
outAlgM ◀ PrfAlgM FixIndM (λ _ . F FixIndM) inFixIndM
  = Λ R. Λ c. λ f. λ y. elimId -(imap c) y.
```

The identity mapping of `F` lifts the identity function  $c : \text{Id } R \ X$  to an identity function  $\text{Id } (F \ R) \ (F \ \text{FixIndM})$ , which is then applied to the argument  $y : F \ R$  to get the desired value of `F FixIndM`.

The proof algebra `outAlgM` induces the constant-time inverse of `inFixIndM`:

```
outFixIndM ◀ FixInd → F FixInd = induction outAlgM.
```

We defined `outFixIndM (inFixIndM x)` as `induction outAlgM (inFixInd x)`, which reduces to `outAlgM -c (induction outAlgM) x` in a constant number of steps (`indHom`). Because `outAlgM -c` erases to  $\lambda f. \lambda y. y$ , it follows that `outFixIndM` computes an inverse of `inFixIndM` in a constant number of beta-reductions:

```
lambek1 ◀ Π x : F FixInd. outFixIndM (inFixIndM x) ≃ x = λ x. β.
```

Furthermore, we show that `outFixIndM` is a post-inverse:

```
lambek2 ◀ Π x : FixIndM. inFixIndM (outFixIndM x) ≃ x
  = λ x. induction (Λ R. Λ c. λ ih. λ fr. β) x.
```

This direction requires us to invoke induction to “pattern match” on the argument value to get  $x := \text{inFixIndM } y$  for some value  $y$  of type `F FixIndM`. Then, `inFixIndM (outFixIndM (inFixIndM y))`  $\simeq$  `inFixIndM y` because the inner term `outFixIndM (inFixIndM y)` is just  $y$  by beta reduction (`lambek1`).

## 6 Examples

In this section, we demonstrate the utility of our derivations on three examples. First, we present a detailed implementation of natural numbers with a constant-time predecessor function. Second, we show examples of infinitary datatypes. Finally, we give an example of a datatype arising as a least fixed point of a scheme that is not a functor, but has an identity mapping.

### 6.1 Natural Numbers with Constant-Time Predecessor

Natural numbers arise as a least fixed point of the functor `NF`:

```
NF ◀ * → * = λ X. Unit + X.
```

```
nfmap ◀ Functor NF = <.>
```

Because every functor induces an identity mapping, we can use our framework to define the natural numbers as follows:

```
nfimap ◀ IdMapping NF = fm2im nfmap.
```

```
Nat ◀ * = FixInd NF nfimap.
```

```
zero ◀ Nat = inFixIndM (in1 unit).
```

```
suc ◀ Nat → Nat = λ n. inFixIndM (in2 n).
```

If we assume that injections `in1` and `in2` erase to  $\lambda a. \lambda i. \lambda j. i a$  and  $\lambda a. \lambda i. \lambda j. j a$ , then the natural number constructors have the following erasures:

```
zero ≈ λ alg. (alg (λ f. (f alg)) (λ i. λ j. (i (λ x. x))))
```

```
suc n ≈ λ alg. (alg (λ f. (f alg)) (λ i. λ j. (j n)))
```

Intuitively, Mendler-style numerals have a constant-time predecessor because every natural number `suc n` contains the previous natural `n` as its direct subpart (which is not true for the Church encoding).

We implement the predecessor for `Nat` in terms of the generic constant-time destructor `outFixIndM`:

```
pred ◀ Nat → Nat = λ n. case (outFixIndM n) (λ _. zero) (λ m. m).
```

Because elimination of disjoint sums (`case`) and `outFixIndM` are both constant-time operations, `pred` is also a constant-time function and its correctness is immediate (i.e., by beta-reduction):

```
predSuc ◀ Π n : Nat. pred (suc n) ≈ n = λ n. β.
```

We also show that the usual “flat” induction principle can be derived from our generic induction principle (`induction`) by dependent elimination of `NF`:

```
indNat ◀ ∀ P : Nat → *. (Π n : Nat. P n → P (suc n)) → P zero
→ Π n : Nat. P n = λ P. λ s. λ z. λ n. induction
  (λ R. λ c. λ ih. λ v. case v (λ u. ρ (etaUnit u) - z)
    (λ r. s (elimId -c r) (ih r))) n.
```

## 6.2 Infinitary Trees

In Agda, we can give the following inductive definition of infinitary trees:

```
data ITree : Set where
  node : (Nat → Unit + ITree) → ITree
```

`ITree` is a least fixed point of functor  $\text{IF } X := \text{Nat} \rightarrow \text{Unit} + X$ . In Cedille, we can implement a functorial function lifting for `IF`:

```
itfmap ◀ ∀ X Y : *. (X → Y) → IF X → IF Y
= λ f. λ t. λ n. case (t n) (λ u. in1 u) (λ x. in2 (f x)).
```

To our best knowledge, it is impossible to prove that `itfmap` satisfies the functorial laws without functional extensionality (which is unavailable in Cedille). However, it is possible to implement an *identity mapping* for the scheme `IF`:

```
itimap ◀ IdMapping IF
= λ c. pair (λ x. λ n. elimId -(nfimap -c) (x n)) β.
```

The first element of a pair erases to  $\lambda x. \lambda n. x n$ , which is  $\lambda x. x$  by the eta law. Because we showed that `IF` has an identity mapping, our generic development induces the datatype `ITree` with its constructor, destructor, and induction principle.

```
ITree ◀ * = FixIndM IF itimap.
```

```
inode ◀ (Nat → Unit + ITree) → ITree = λ f. inFixIndM f.
```

Below, the specialized induction is phrased in terms of the “empty tree” `iempty`, which acts as a base case (`projR` “projects” a tree from a disjoint sum or returns `iempty`):

```
iempty ◀ ITree = inode (λ _ . in1 unit).
```

```
indITree ◀ ∀ P : ITree → *. P iempty →
(Π f : Nat → Unit + ITree. (Π n : Nat. P (projR (f n))
→ P (inode f)) → Π t : ITree. P t = <.>
```

Next, let us look at another variant of infinitary datatypes in Agda:

```
data PTree : Set where
  pnode : ((PTree → Bool) → Unit + PTree) → PTree
```

This definition will be rejected by Agda (and Coq) since it arises as a least fixed point of the scheme  $\text{PF } X := \text{Unit} + ((X \rightarrow \text{Bool}) \rightarrow X) \rightarrow X$ , which is positive but not strictly positive. The definition is rejected because it is currently unclear if non-strict definitions are sound in Agda. For the Coq setting, there is a proof by Coquand and Paulin that non-strict positivity combined with an impredicative universe and a predicative universe hierarchy leads to inconsistency [14]. In Cedille, we can implement an identity mapping for the scheme `PF` in a similar fashion as the previously discussed `UF`. Hence, the datatype induced by `PF` exists in the type theory of Cedille.

### 6.3 Unbalanced Trees

Consider the following definition of “unbalanced” binary trees in Agda:

```
data UTree : Set where
  leaf : Bool → UTree
  node : (b1 : UTree) → (b2 : UTree) → b1 ≠ b2 → UTree
```

The datatype `UTree` arises as a least fixed point of the following scheme:

$$\text{UF} \triangleleft \star \rightarrow \star = \lambda X. \text{Bool} + (\Sigma x_1 : X. \Sigma x_2 : X. x_1 \neq x_2).$$

Because the elements  $x_1$  and  $x_2$  must be different, lifting an arbitrary function  $X \rightarrow Y$  to  $\text{UF } X \rightarrow \text{UF } Y$  is impossible. Hence, the scheme `UF` is not a functor.

However, we can show that `UF` has an identity mapping. We start by producing a function  $\text{UF } X \rightarrow \text{UF } Y$  from an identity  $\text{Id } X Y$ :

```
uimap' ◀ ∀ X Y : ★. Π i : Id X Y. UF X → UF Y = λ i. λ uf.
  case uf (λ u. in1 u)
    (λ u. in2 (pair (elimId -i (π1 u))
                  (pair (elimId -i (π1 (π2 u))) (π2 (π2 u))))))
```

We prove that `uimap' -i` is extensionally an identity function:

$$\text{uimP} \triangleleft \forall X Y : \star. \Pi i : \text{Id } X Y. \Pi u : \text{UF } X. u \simeq \text{uimap}' i u = \langle \dots \rangle$$

This is enough to derive an identity mapping for `UF` by using the previously implemented combinator `intrId`:

$$\text{uimap} \triangleleft \text{IdMapping UF} = \lambda i. \text{intrId (uimap}' i) (\text{uimP } i).$$

Therefore, we conclude that the datatype of unbalanced trees exists in Cedille and can be defined as a least fixed point of the scheme `UF`:

$$\text{UTree} \triangleleft \star = \text{FixIndM UF uimap}.$$

The specialized constructors, induction principle, and a destructor function for `UTree` are easily derived from their generic counterparts (`inFixIndM`, `induction`, `outFixIndM`).

## 7 Related Work

Pfenning and Paulin-Mohring show how to model inductive datatypes using impredicative encodings in the Calculus of Constructions (CC) [15]. Because induction is not provable in the CC, the induction principles are generated and added as axioms. This approach was adopted by initial versions of the Coq proof assistant, but later Coq switched to the Calculus of Inductive Constructions (CIC), which has built-in inductive datatypes.

Delaware et al. derived induction for impredicative lambda-encodings in Coq as a part of their framework for modular definitions and proofs (using the à la

carte technique [16]). They showed that a value  $v : \mathbf{Fix}\ F$  is inductive if it is accompanied by a proof of the universal property of folds [17].

Similarly, Torrini introduced the *predicatisation* technique, reducing dependent induction to proofs that only rely on non-dependent Mendler induction (by requiring the inductive argument to satisfy an extra predicatisation hypothesis) [18].

Traytel et al. present a framework for constructing (co)datatypes in HOL [19, 20]. The main ingredient is a notion of a bounded natural functor (BNF), or a binary functor with additional structure. BNFs are closed under composition and fixed points, which enables support for both mutual and nested (co)recursion with mixed combinations of datatypes and codatatypes. The authors developed a package that can generate (co)datatypes with their associated proof-principles from user specifications (including custom bounded natural functors). In contrast, our approach provides a single generic derivation of induction within the theory of Cedille, but does not address codatatypes. It would be interesting to further investigate the exact relationship between schemes with identity mappings and BNFs.

Church encodings are typeable in System F and represent datatypes as their own iterators. Parigot proved that the lower bound of the predecessor function for Church numerals has linear time complexity [4].

Parigot designed an impredicative lambda-encoding that is typeable in System  $F_\omega$  with positive-recursive type definitions. The encoding identifies datatypes with their own recursors, allowing constant time destructors to be defined, but the drawback is that the representation of a natural number  $n$  is exponential in the call-by-value setting [5].

The Stump-Fu encoding is also typeable in System  $F_\omega$  with positive-recursive type definitions. It improves upon the Parigot representation by requiring only quadratic space, and it also supports constant-time destructors [6].

## 8 Conclusions and Future Work

In this work, we showed that the Calculus of Dependent Lambda Eliminations is a compact pure type theory that allows a general class of Mendler-style lambda-encoded inductive datatypes to be defined as least fixed points of schemes with identity mappings. We also gave a generic derivation of induction and implemented a constant-time destructor for these datatypes. We used our development to give the first example (to the best of our knowledge) of lambda-encoded natural numbers with: provable induction, a constant-time predecessor function, and a linear size (in the numeral  $n$ ) term representation. Our formal development is around 700 lines of Cedille code.

For future work, we plan to explore coinductive definitions and to use the categorical model of Mendler-style datatypes to investigate isomorphisms and inductive-recursive datatypes in Cedille [12].

Additionally, we are investigating approaches to generic “proof reuse” in Cedille. The key idea is that implicit products allow indexed and non-indexed



datatypes to be represented by the same pure lambda term. For example, this allows the proof of associativity of `append` for lists to be reused as the proof of associativity of `append` for vectors [11].

**Acknowledgments.** The first author is thankful to Anna, Albert, and Eldar for all the joy and support. Authors are thankful to Larry Diehl for proof reading and numerous grammatical adjustments. We gratefully acknowledge NSF support under award 1524519, and DoD support under award FA9550-16-1-0082 (MURI program).

## References

1. Geuvers, H.: Induction Is Not Derivable in Second Order Dependent Type Theory. In: *Typed Lambda Calculi and Applications (TLCA)*. (2001) 166–181
2. Stump, A.: The Calculus of Dependent Lambda Eliminations. *Journal of Functional Programming* **27** e14
3. Firsov, D., Stump, A.: Generic derivation of induction for impredicative encodings in cedille. In: *Proceedings of the 7th ACM SIGPLAN International Conference on Certified Programs and Proofs. CPP 2018, New York, NY, USA, ACM* (2018) 215–227
4. Parigot, M.: On the representation of data in lambda-calculus. In Börger, E., Bünning, H.K., Richter, M.M., eds.: *CSL '89, Springer Berlin Heidelberg* (1990) 309–321
5. Parigot, M. In: *Programming with proofs: A second order type theory*. Springer Berlin Heidelberg, Berlin, Heidelberg (1988) 145–159
6. Stump, A., Fu, P.: Efficiency of lambda-encodings in total type theory. *Journal of Functional Programming* **26** (2016)
7. Stump, A.: From Realizability to Induction via Dependent Intersection. *Ann. Pure Appl. Logic* (2018) to appear.
8. Wells, J.B.: Typability and type checking in system F are equivalent and undecidable. *Ann. Pure Appl. Logic* **98**(1-3) (1999) 111–156
9. Kopylov, A.: Dependent intersection: A new way of defining records in type theory. In: *18th IEEE Symposium on Logic in Computer Science (LICS)*. (2003) 86–95
10. Miquel, A.: The Implicit Calculus of Constructions Extending Pure Type Systems with an Intersection Type Binder and Subtyping. In Abramsky, S., ed.: *Typed Lambda Calculi and Applications (TLCA)*. (2001) 344–359
11. Diehl, L., Firsov, D., Stump, A.: Generic zero-cost reuse for dependent types. *CoRR* **abs/1803.08150** (2018)
12. Uustalu, T., Vene, V.: Mendler-style inductive types, categorically. *Nordic J. of Computing* **6**(3) (September 1999) 343–361
13. Wadler, P.: Recursive types for free! (1990)
14. Coquand, T., Paulin, C.: Inductively defined types. In Martin-Löf, P., Mints, G., eds.: *COLOG-88, Berlin, Heidelberg, Springer Berlin Heidelberg* (1990) 50–66
15. Pfenning, F., Paulin-Mohring, C.: Inductively defined types in the Calculus of Constructions. In Main, M., Melton, A., Mislove, M., Schmidt, D., eds.: *Proceedings of the Fifth Conference on the Mathematical Foundations of Programming Semantics, Tulane University, New Orleans, Louisiana, Springer-Verlag LNCS 442* (March 1989) 209–228

16. Swierstra, W.: Data types à la carte. *J. Funct. Program.* **18**(4) (July 2008) 423–436
17. Delaware, B., d. S. Oliveira, B.C., Schrijvers, T.: Meta-theory à la carte. In: *Proceedings of the 40th Annual ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages*. POPL '13, New York, NY, USA, ACM (2013) 207–218
18. Torrini, P.: Modular dependent induction in coq, mendler-style. In Blanchette, J.C., Merz, S., eds.: *Interactive Theorem Proving*, Springer International Publishing (2016) 409–424
19. Traytel, D., Popescu, A., Blanchette, J.C.: Foundational, Compositional (Co)datatypes for Higher-Order Logic: Category Theory Applied to Theorem Proving. In: *Proceedings of the 27th Annual IEEE Symposium on Logic in Computer Science, LICS 2012, Dubrovnik, Croatia, June 25-28, 2012*, IEEE Computer Society (2012) 596–605
20. Biendarra, J., Blanchette, J.C., Bouzy, A., Desharnais, M., Fleury, M., Hölzl, J., Kuncar, O., Lochbihler, A., Meier, F., Panny, L., Popescu, A., Sternagel, C., Thiemann, R., Traytel, D.: Foundational (Co)datatypes and (Co)recursion for Higher-Order Logic. In Dixon, C., Finger, M., eds.: *Frontiers of Combining Systems - 11th International Symposium, FroCoS 2017, Brasília, Brazil, September 27-29, 2017*, Proceedings. Volume 10483 of *Lecture Notes in Computer Science.*, Springer (2017) 3–21